

Comprehensions in Mozart: Syntax

This paper aims at explaining the complete syntax of comprehensions in Mozart2. There are two kinds of comprehensions implemented for the moment: list and record comprehensions. There are two main differences between the two later. First, list comprehensions return list(s) while record comprehensions return record(s). Second, there are more functionalities available with list comprehensions.

1 List comprehensions

A list comprehension allows the creation of lists in an easy and flexible way. In this section, we explain their complete syntax incrementally.

An example of a very simple list comprehension is:

```
[A+1 suchthat A in 1..5] % = [2 3 4 5 6]
```

Let us analyze this comprehension. The first thing that interests us is the **suchthat A in 1..5**. This means that the list comprehension creates a new variable named **A** and that this variable will take values from 1 to 5. The output is specified by the first part, **A+1**. It means that we just add the current value of **A** incremented by 1 to the output list. This expression can be anything.

We will start from this simple syntax and, step by step, add functionalities. All functionalities are compatible with each other, even if examples do not cover all combinations.

1.1 Generator, ranger and layer

A *generator* is the specification of a structure on which the comprehension iterates. In our example, **1..5** is a generator, it iterates from 1 to 5.

The *ranger* is the structure that is assigned to the current value of its corresponding generator at each iteration. In our example, **A** is the ranger of the generator we just saw.

A *layer* is the aggregation of a ranger and its generator.

There are five kinds of generators. Here are the descriptions of the first four (examples follow):

Integer generator: `in Low..High ; Step`

Goes from integers **Low** to **High** by integer step of **Step**. The step is optional, default is 1. **Low** must be smaller or equal to **High**.

C-style generator: `in (Init ; Condition ; Next)`

Initiate its ranger to **Init**, iterates while **Condition** evaluates to true and updates the ranger to **Next** at the end of each iteration. The parenthesis are optional. The condition is optional, default is **true** (this implies that the generator never stops iterating by itself).

List generator: `in List`

List can be a variable containing a list or the definition of new list. The generator iterates over all the elements of the list in the same order as they appear in the list. The list can be a stream.

Function generator: `from Function`

Function can be a variable containing a function or the definition of a unnamed function. In both cases, the function does not take any arguments. The generator never stops. At each iteration, the current value of the generator is the result of calling **Function**.

```

% Integer generator
[A suchthat A in 1..5 ; 2] % [1 3 5]
[A suchthat A in 1..5 ; 1] % [1 2 3 4 5]
[A suchthat A in 1..5] % [1 2 3 4 5]
% C-style generator
[A suchthat A in 1 ; A<6 ; A+1] % [1 2 3 4 5]
[A suchthat A in ( 1 ; A<6 ; A+1 )] % [1 2 3 4 5]
[A suchthat A in 1 ; A+1] % [1 2 3 4 5 ...] (never stops)
% List generator
L = [1 2 3]
[A suchthat A in L] % [1 2 3]
[A suchthat A in [1 2 3]] % [1 2 3]
[A suchthat A in [1 [2] 3]] % [1 [2] 3]
% Function generator
Fun = fun{$} 1 end
[A suchthat A from Fun] % [1 1 1 1 ...] (never stops)
[A suchthat A from fun{$} 1 end] % [1 1 1 1 ...] (never stops)

```

The fifth kind of generator, record generators, is a bit more complicated. It is comparable to list generators but record generators are more general. A record generator has the following form:

```
F:V in Record
```

As for lists, **Record** can be a variable containing a record or the definition of a new record. If it is a variable, then we have no way to know in advance what kind of generator it is. That is why we need a way to differentiate them. The ranger contains a feature **F** and a value **V**. This is how we differentiate record from list generators. Note that this is the reason why the function generator uses the **from** keyword and not **in**.

Iterating over a record with a feature and a value means that at every iteration, the feature is assigned to the current feature and the value is assigned to the current value. Such rangers allow having all the field information available.

The iteration goes through all fields of the record in the same order as its arity.

Here are some examples of list comprehensions with record generators:

```

R = r(a:1 b:r(1 2))
[A suchthat _:A in 1#2#3] % [1 2 3]
[F A suchthat F:A in R] % [a b]#[1 r(1 2)]
[F suchthat F:_ in 1#2#3 if F > 1] % [2 3]
[F#A suchthat F:A in r(a:1 b:r(1 2))] % [a#1 b#r(1 2)]
[F#A if F == b suchthat F:A in R] % [b#r(1 2)]

```

1.2 Multi layer

Now that all kinds of generators have been specified, we can move on to specifying several layers. One specifies several layers as follows:

```

[... suchthat Layer_1 ... Layer_N]
% Example
[[A B] suchthat A in 1..3 B in [a b]] % [[1 a] [2 b]]

```

Layers are traversed simultaneously. It means that a layer stops iterating when its iteration has reached its end or once at least one of its neighbor layers stops iterating. So the *smallest* generator (with the smallest number of iterations) decides when to stop iterating. In our example above, B is the range of the smallest generator and will take the values a then b. A will follow B and take the values 1 then 2, never 3.

Thanks to this functionality, function generators can be stopped as well as C-style generators without a condition.

1.3 Multi level

A *level* is the aggregation of one or more layers traversed simultaneously (inside the same *suchthat*) with an optional condition. A list comprehension contains at least one level.

Levels are like nested loops. It means that for each element of the main loop, the nested loop will run all its iterations. Here is an example with two levels:

```
L = [4 7]
[[A 2*B] suchthat A in L suchthat B in 3..5] % [[4 6] [4 8] [7 6] [7 8]]
```

In addition to all its layers, a level can have a condition. The latter is always at the end of the level and is delimited by the keyword *if*. The condition is a boolean expression. When the condition evaluates to true, the iteration is accepted and the next level is called if it exists. If it does not then an element is added to the output list. When the condition evaluates to false, the current iteration of the current level is skipped. Here are some examples:

```
L = [1 2 4 7]
[[A 2*B] suchthat A in L if A>6 suchthat B in 3..5] % [[7 6] [7 8] [7 10]]
[A+B suchthat A in L suchthat B in 3..5 if A>B] % [7 10 11 12]
[A+B suchthat A in L suchthat B in 3..5 if A>B andthen B<5] % [7 10 11]
```

1.4 Multi output

Up to now, we limited ourselves to one output list. List comprehensions allow the result to have more than one output lists. All these output lists are put inside a record. One just specifies one expression by output list as follows:

```
[a b suchthat _ in 1..2] % [a a]#[b b]
```

When features are not specified, they are implicitly created to be the smallest unused integers from one. The label is always '#' so that the record can also be equivalent to a tuple declared using its syntactic sugar.

To specify the feature, one just needs to use a colon to separate the feature from the expression. Here are two examples:

```
[a:a b:b suchthat _ in 1..2] % '#'(a:[a a] b:[b b])
[a 1:b suchthat _ in 1..2] % [b b]#[a a]
```

In addition, each output can have a condition called the *output condition*. This condition allows filtering the elements of the corresponding list. Thanks to this functionality, output lists can have different sizes. Here is an example:

```
[smallerEquals:A if A ≤ 3 bigger:A if A > 3 suchthat A in [3 4 2 8 5 7 6]]
% '#'(smallerEquals:[3 2] bigger:[4 8 5 7 6])
```

The output is a list only when there is one output expression without a feature, otherwise, the result is a record of list(s). So the two following examples are different:

```
[1:A if A < 3 suchthat A in 1..5] % '#'(1:[1 2])
[A if A < 3 suchthat A in 1..5] % [1 2]
```

1.5 Laziness

A level can be specified as lazy. It is done by specified a lazy flag as any other layer (the flag is considered as a layer):

```
declare L in
thread L = [A suchthat lazy A in 1..5] end
% L = _
{List.drop L 1 _}
% L = 1|_
{List.drop L 3 _}
% L = 1|2|3|_
```

As levels are specified as lazy, one must choose wisely which level(s) must be lazy as in the following example:

```
declare L in
thread L = [A#B suchthat lazy A in 1..5 suchthat B in [a b]] end
% L = _
{List.drop L 1 _}
% L = 1#a|1#b|_
{List.drop L 2 _}
% L = 1#a|1#b|_
{List.drop L 3 _}
% L = 1#a|1#b|2#a|2#b|_
```

As the first level is declared as lazy, this level waits for an element to be needed to launch the iteration over its next element. When a value is needed, the first level allows one iteration to execute. This means that the second level can traverse all of its iterations because it is not lazy. That is why in the above example, elements are created by two (the number of iterations of the second level).

1.6 Bodies

As a list comprehension has an arbitrary number of levels, it is comparable to nested loops. This comparison has driven us to allow list comprehensions to have a body. The *body* of a list comprehension is a statement that will be executed (just) before each time the comprehension adds an element to each output, or when the comprehension tries appending even if output conditions are false. In the comparison with nested loops, the body is the statement inside the *deepest* loop of the nesting. Here is an example:

```
[A suchthat A in 1..5 do {Delay 1000}] % [1 2 3 4 5] after 5 seconds
```

2 Record comprehensions

The principle is basically the same as for list comprehensions except that instead of returning a list (or several ones) record comprehensions return a record (or several ones). The idea is to keep the same shape, the same arity as the input record. For this reason, record comprehensions only take one record as input. In other words, we could say that record comprehensions are restricted to one layer and one level. On the other hand we still keep the multi output with individual conditions. The level condition can also be specified.

Because record comprehensions output a record, or records, similar to the input, we decided not to allow several levels. If one wants to use several layers then it would imply that all input records have similar arities. For these reasons, we decided to restrict record comprehensions to one level and one layer.

As record comprehensions use a syntax very similar to the one of list comprehensions, we need to differentiate them. This is done by using parenthesis instead of square brackets to delimit the comprehension.

Here are some examples:

(A suchthat _:A in r(a:2 b:3 1))	% r(1:1 a:2 b:3)
(A+1 suchthat _:A in r(a:2 b:3 1))	% r(1:2 a:3 b:4)
(@C suchthat _:A in r(a:2 b:3 1) do C := A)	% r(1:2 a:3 b:4)
(F suchthat F:_ in r(a b c))	% r(1 2 3)
(F+1 suchthat F:_ in r(a b c))	% r(2 3 4)
(F#A suchthat F:A in r(a:1 b:r(1 2)))	% r(a:a#1 b:b#r(1 2))
(A suchthat _:A in r(a:1 b:r(1 2)) if F == b)	% r(b:r(1 2))
(f:F a:A suchthat F:A in r(a b))	% '#'(f:r(1 2) a:r(a b))
(A if F == 2 A suchthat F:A in r(a b c))	% r(b)#r(a b c)